
AWS Signature Version 4 signing library for Elixir. Small, focused, and easy to use.
Features
	Req Plugin: Automatic request signing for the Req HTTP client
	Clean API: Modern credential struct-based interface
	Session Tokens: Full support for temporary AWS credentials (STS)
	Flexible Output: Choose between list, map, or Req-compatible header formats
	Smart Defaults: Auto-detect service and region from AWS URLs
	Presigned URLs: Configurable expiration times (15 min default, up to 7 days)
	Streaming Support: Unsigned payload option for large file uploads
	Backward Compatible: Maintains compatibility with the original aws_auth API

Installation
Add ex_aws_auth to your dependencies in mix.exs:
def deps do
  [
    {:ex_aws_auth, "~> 1.3"},
    {:req, "~> 0.5"}  # Optional, for Req plugin
  ]
end
Quick Start
With Req (Recommended)
The easiest way to use this library is with the Req plugin:
# Load credentials from environment (AWS_ACCESS_KEY_ID, AWS_SECRET_ACCESS_KEY, etc.)
creds = AWSAuth.Credentials.from_env()

# Make authenticated requests
Req.new(url: "https://s3.amazonaws.com/my-bucket/file.txt")
|> AWSAuth.Req.attach(credentials: creds, service: "s3")
|> Req.get!()

# POST with body
Req.new(url: "https://bedrock-runtime.us-east-1.amazonaws.com/model/my-model/invoke")
|> AWSAuth.Req.attach(credentials: creds, service: "bedrock")
|> Req.post!(json: %{prompt: "Hello"})
Manual Signing
For manual control over request signing:
# Create credentials
creds = %AWSAuth.Credentials{
  access_key_id: "AKIAIOSFODNN7EXAMPLE",
  secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  region: "us-east-1"
}

# Sign a URL (for presigned URLs)
signed_url = AWSAuth.sign_url(
  creds,
  "GET",
  "https://s3.amazonaws.com/my-bucket/file.txt",
  "s3"
)

# Sign request headers (for API calls)
headers = AWSAuth.sign_authorization_header(
  creds,
  "POST",
  "https://bedrock-runtime.us-east-1.amazonaws.com/model/my-model/invoke",
  "bedrock",
  headers: %{"content-type" => "application/json"},
  payload: Jason.encode!(%{prompt: "Hello"})
)
Credentials
From Environment Variables
# Reads AWS_ACCESS_KEY_ID, AWS_SECRET_ACCESS_KEY, AWS_SESSION_TOKEN, AWS_REGION
creds = AWSAuth.Credentials.from_env()
Manual Creation
creds = %AWSAuth.Credentials{
  access_key_id: "AKIAIOSFODNN7EXAMPLE",
  secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  session_token: "FwoGZXIvYXdzEBYa...",  # Optional, for STS temporary credentials
  region: "us-east-1"
}
From Map or Keyword List
creds = AWSAuth.Credentials.from_map(%{
  access_key_id: "...",
  secret_access_key: "...",
  region: "us-west-2"
})
Advanced Usage
Auto-Detection of Service and Region
Skip specifying service and region when using standard AWS URLs:
creds = AWSAuth.Credentials.from_env()

# Service and region detected from URL
signed_url = AWSAuth.sign_url(
  creds,
  "GET",
  "https://s3.us-west-2.amazonaws.com/my-bucket/file.txt",
  nil  # nil = auto-detect
)

# Works with any AWS service
headers = AWSAuth.sign_authorization_header(
  creds,
  "POST",
  "https://bedrock-runtime.us-east-1.amazonaws.com/model/invoke",
  nil,
  payload: body
)
Presigned URLs with Custom Expiration
Control how long presigned URLs remain valid:
# Default: 15 minutes
signed_url = AWSAuth.sign_url(creds, "GET", url, "s3")

# Custom expiration: 1 hour
signed_url = AWSAuth.sign_url(creds, "GET", url, "s3", expires_in: 3600)

# Maximum: 7 days
signed_url = AWSAuth.sign_url(creds, "GET", url, "s3", expires_in: 604_800)
Streaming Uploads (Unsigned Payload)
For large file uploads where the payload isn't known upfront:
# Use :unsigned for streaming
headers = AWSAuth.sign_authorization_header(
  creds,
  "PUT",
  "https://s3.amazonaws.com/bucket/large-file.zip",
  "s3",
  payload: :unsigned
)
Event Stream Signing (Bidirectional Streaming)
Some AWS APIs accept a stream of events from the client over a single
connection (e.g. Amazon Transcribe streaming, Bedrock bidirectional
streaming). Each event is signed individually, and the signatures form a
chain: the first event uses the seed signature from the initial request's
Authorization header, and each subsequent event uses the previous event's
signature.
creds = AWSAuth.Credentials.from_env()

# `seed` is the Signature=... value from the initial request's Authorization header
{frame1, sig1} =
  AWSAuth.EventStream.sign_message(creds, "transcribe", seed, payload1, NaiveDateTime.utc_now())

{frame2, sig2} =
  AWSAuth.EventStream.sign_message(creds, "transcribe", sig1, payload2, NaiveDateTime.utc_now())

# `frame1` / `frame2` are wire-ready event stream messages; send them on the connection.
sign_message/6 encodes the :date header, signs the event, and frames a
complete event stream message (prelude + CRCs + :date and
:chunk-signature headers + payload). For full control over the encoded
header bytes, use the low-level AWSAuth.EventStream.sign_event/7.
Session Tokens (STS Temporary Credentials)
Full support for AWS Security Token Service temporary credentials:
creds = %AWSAuth.Credentials{
  access_key_id: "ASIAIOSFODNN7EXAMPLE",  # Note: Starts with ASIA for temporary creds
  secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  session_token: "FwoGZXIvYXdzEBYaDHhBTEMPLESessionToken123",
  region: "us-east-1"
}

# Automatically includes X-Amz-Security-Token in signatures
Req.new(url: url)
|> AWSAuth.Req.attach(credentials: creds, service: "s3")
|> Req.get!()
Header Format Options
Choose your preferred header format:
# List of tuples (default, works with most HTTP clients)
headers = AWSAuth.sign_authorization_header(creds, method, url, service,
  return_format: :list
)
# => [{"authorization", "AWS4-HMAC-SHA256 ..."}, {"x-amz-date", "..."}]

# Map (convenient for merging)
headers = AWSAuth.sign_authorization_header(creds, method, url, service,
  return_format: :map
)
# => %{"authorization" => "AWS4-HMAC-SHA256 ...", "x-amz-date" => "..."}

# Req format (for use with Req)
headers = AWSAuth.sign_authorization_header(creds, method, url, service,
  return_format: :req
)
# => %{"authorization" => ["AWS4-HMAC-SHA256 ..."], "x-amz-date" => ["..."]}
Advanced Options
Fine-tune signing behavior for edge cases:
# Exclude specific headers from signing
headers = AWSAuth.sign_authorization_header(creds, "POST", url, "service",
  unsigned_headers: ["user-agent", "accept-encoding"]
)

# Disable URI path escaping (for pre-encoded paths)
signed_url = AWSAuth.sign_url(creds, "GET", url, "s3",
  uri_escape_path: false
)

# Omit content checksum header (rare)
headers = AWSAuth.sign_authorization_header(creds, "POST", url, "service",
  apply_checksum_header: false
)

# Override region from credentials
signed_url = AWSAuth.sign_url(creds, "GET", url, "s3",
  region: "eu-west-1"
)
Legacy API (Backward Compatibility)
The original aws_auth API is still supported for backward compatibility:
# Old-style URL signing (still works)
signed_url = AWSAuth.sign_url(
  "AKIAIOSFODNN7EXAMPLE",
  "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  "GET",
  "https://examplebucket.s3.amazonaws.com/test.txt",
  "us-east-1",
  "s3"
)

# Old-style header signing (still works)
auth_header = AWSAuth.sign_authorization_header(
  "AKIAIOSFODNN7EXAMPLE",
  "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  "PUT",
  "https://examplebucket.s3.amazonaws.com/test.txt",
  "us-east-1",
  "s3",
  %{"x-amz-storage-class" => "REDUCED_REDUNDANCY"},
  "file contents"
)
Note: While the legacy API is fully supported, we recommend using the new Credentials-based API or Req plugin for new projects.
Why ex_aws_auth?
The original aws_auth package was created by Bryan Joseph. Unfortunately, it is no longer actively maintained and has become incompatible with modern versions of Erlang/OTP.
Rodrigo Zampieri Castilho forked it and brought it up to date for OTP 27, but his fork cannot be included as a dependency in packages published on Hex.
This package, ex_aws_auth, is published on Hex to make this essential AWS signing functionality available to the broader Elixir community, with modern improvements and active maintenance.
Full credit and thanks go to Bryan Joseph for creating the original library and Rodrigo Zampieri Castilho for maintaining compatibility with modern OTP versions.
Documentation
Full documentation is available on HexDocs.
License
Apache License 2.0 - See LICENSE file for details.


  

    CHANGELOG

[1.4.1] - 2026-06-15
Added
	Request-side helpers that complete the event-stream support added in 1.4.0:	AWSAuth.EventStream.encode_string_header/2 — type-7 string header encoder
(for application headers like :content-type/:event-type/:message-type).
	:streaming_events payload option for sign_authorization_header/5, which
seeds an event-stream request with
x-amz-content-sha256: STREAMING-AWS4-HMAC-SHA256-EVENTS.




[1.4.0] - 2026-06-15
Added
	Event stream signing for bidirectional streaming APIs (AWSAuth.EventStream)	sign_message/6 signs an event and returns a wire-ready event stream frame
(prelude + CRCs + :date/:chunk-signature headers + payload), plus the
signature to chain into the next event
	sign_event/7 low-level signer matching the canonical
AWS4-HMAC-SHA256-PAYLOAD algorithm
	Helpers for encoding event stream headers and framing messages
	Used by APIs such as Amazon Transcribe streaming and Bedrock bidirectional
streaming



Changed
	Minimum Elixir is now ~> 1.15 (was ~> 1.14). The optional req
dependency pulls in finch, which dropped Elixir 1.14 support as of 0.22.
	Updated dependencies to latest and pruned stale unused lockfile entries
inherited from the original aws_auth.
	CI now tests Elixir 1.20 / OTP 29 as the primary target.


[1.3.1] - 2025-01-18
Changed
	Updated Elixir to 1.19.2 and Erlang to 28.1.1
	Updated ex_doc to 0.39.1
	Updated CI test matrix to use latest tooling versions


[1.3.0] - 2025-01-17
Added
	Service and region auto-detection from AWS URLs
	Pass nil as service to automatically detect from URL (e.g., https://s3.us-west-2.amazonaws.com)
	Region is auto-detected from URL if not provided in credentials or options
	Supports standard AWS service patterns (service.region.amazonaws.com)


	Presigned URL expiration control
	:expires_in option for customizing presigned URL validity period
	Default: 900 seconds (15 minutes), maximum: 604800 seconds (7 days)
	Changed default expiration from 86400 (24 hours) to 900 (15 minutes) to match AWS SDK conventions


	Unsigned payload support for streaming
	:unsigned atom can now be passed as payload parameter
	Useful for large file uploads where content isn't known upfront
	Automatically sets payload to "UNSIGNED-PAYLOAD" in signature


	Advanced signing options
	:unsigned_headers - Exclude specific headers from signing (e.g., ["user-agent"])
	:uri_escape_path - Control URI path escaping (default: true)
	:apply_checksum_header - Control inclusion of x-amz-content-sha256 header (default: true)



Changed
	Breaking: Default presigned URL expiration reduced from 24 hours to 15 minutes	This aligns with AWS SDK defaults and security best practices
	Explicitly set :expires_in if you need longer expiration times


	Internal header filtering now accepts a custom unsigned-headers list
	Internal canonical-request building now supports URI path escaping control
	All legacy API signatures remain fully backward compatible with new optional parameters


[1.2.0] - 2025-01-17
Added
	AWSAuth.Credentials struct for cleaner credential management	from_env/0 to load credentials from environment variables
	from_map/1 to create from maps or keyword lists


	Credential struct overloads for simpler signing	sign_url(credentials, method, url, service, opts)
	sign_authorization_header(credentials, method, url, service, headers, payload, opts)


	return_format option for flexible header formats	:list (default) - Returns list of tuples [{header, value}]
	:map - Returns map %{header => value}
	:req - Returns Req-compatible format %{header => [value]}


	AWSAuth.Req plugin module for seamless Req integration	Automatically signs Req requests with AWS Signature V4
	Handles header format conversion transparently
	Supports all credential types (long-term, STS temp credentials)


	Optional Req and Jason dependencies for plugin support

Changed
	Timestamp parameter now optional in credential struct APIs (defaults to current time)
	All new APIs maintain 100% backward compatibility with existing function signatures


[1.1.0] - 2025-01-17
Added
	Session token support for temporary AWS credentials (STS AssumeRole)	New optional session_token parameter for sign_url/10 and sign_authorization_header/10
	Automatically adds X-Amz-Security-Token header for authorization header signing
	Automatically adds X-Amz-Security-Token query parameter for URL signing


	Header filtering to remove unsignable headers (x-amzn-trace-id)	Prevents AWS infrastructure trace headers from breaking signatures


	Header value normalization	Collapses multiple consecutive spaces to single space per AWS Sig V4 spec


	Query parameter validation	Rejects list keys/values with clear error messages



Changed
	Headers are now filtered and normalized before signing in all signing methods
	Query parameters are validated before canonical request generation


[1.0.1] - 2025-01-16
Changed
	Updated to Elixir 1.19.0 and OTP 28.1 in CI/CD pipeline
	Updated .tool-versions to Elixir 1.19.0/OTP 28.1


[1.0.0] - 2025-01-15
Initial Release
This is a fork of the original aws_auth package by Bryan Joseph, incorporating OTP 27 compatibility fixes from Rodrigo Zampieri Castilho's fork. Published as ex_aws_auth to make this maintained version available on Hex.
Changed
	Updated minimum Elixir requirement to ~> 1.14
	Modernized all dependencies (ex_doc ~> 0.34, credo ~> 1.7, excoveralls ~> 0.18)
	Migrated from deprecated Mix.Config to import Config

Added
	Quokka ~> 2.11 formatter plugin for code quality
	dialyxir ~> 1.4 for static analysis
	Comprehensive .formatter.exs configuration


Historical Changelog (from original aws_auth package)
[0.6.1]
Fixed
	Correctly handle NaiveDateTimes with ms precision (thanks to @radar)

[0.6.0]
Changed
	Requires Elixir 1.3 or higher

Fixed
	Removed timex dependency and using Elixir's built in datetime functions (thanks to @radar)

[0.5.1]
Fixed
	Use Timex.DateTime.now, rather than Timex.DateTime.today (thanks to @radar)

[0.5.0]
Fixed
	x-amz-date using Date instead of DateTime (thanks to @radar)

Changed
	Dependency updates (thanks to @radar)

[0.4.0]
Fixed
	Signing works for more than just S3 from @kenta-aktsk

Changed
	headers params for sign_url and sign_authorization_header now expects a map instead of a Dict

Signs urls or authentication headers for use with AWS requests.
New in 1.2.0
This version adds convenient new APIs while maintaining 100% backward compatibility:
	AWSAuth.Credentials struct for cleaner credential management
	Credential struct overloads for sign_url and sign_authorization_header
	Optional return_format option (:list, :map, :req) for flexible header formats
	AWSAuth.Req plugin for seamless Req integration

Migration Guide
Old API (still fully supported)
AWSAuth.sign_authorization_header(
  access_key,
  secret_key,
  "POST",
  url,
  "us-east-1",
  "bedrock",
  headers,
  body,
  NaiveDateTime.utc_now(),
  session_token
)
New credential struct API
creds = %AWSAuth.Credentials{
  access_key_id: access_key,
  secret_access_key: secret_key,
  session_token: session_token,
  region: "us-east-1"
}

# Or load from environment
creds = AWSAuth.Credentials.from_env()

# Simpler function call
AWSAuth.sign_authorization_header(
  creds,
  "POST",
  url,
  "bedrock",
  headers,
  body
)
With Req plugin
creds = AWSAuth.Credentials.from_env()

Req.new(url: url, method: :post, body: body)
|> AWSAuth.Req.attach(credentials: creds, service: "bedrock")
|> Req.request()

        sign_authorization_header(creds, http_method, url, service, opts \\ [])

      


        Signs request headers using AWS Signature V4 with a Credentials struct.

        sign_url(creds, http_method, url, service, opts \\ [])

      


        Signs a URL using AWS Signature V4 with a Credentials struct.

        AWSAuth.sign_url(access_key, secret_key, http_method, url, region, service, headers, request_time, payload, session_token)

Signs request headers using AWS Signature V4 with a Credentials struct.
This is a convenience overload that accepts an AWSAuth.Credentials struct.
Parameters
	credentials - AWSAuth.Credentials struct with AWS credentials
	http_method - HTTP method as string ("GET", "POST", etc)
	url - The AWS URL to sign
	service - AWS service name (e.g., "s3", "bedrock"). Can be nil to auto-detect from URL.
	opts - Keyword list of options:	:headers - Request headers as map (default: %{})
	:payload - Request body (default: ""), :unsigned for streaming
uploads, or :streaming_events to seed an event-stream request
(x-amz-content-sha256: STREAMING-AWS4-HMAC-SHA256-EVENTS; see AWSAuth.EventStream)
	:timestamp - NaiveDateTime for signing (default: current time)
	:region - Override region from credentials (auto-detected from URL if not provided)
	:return_format - Return format (:list, :map, :req) (default: :list)
	:unsigned_headers - List of header names to exclude from signing (default: [])
	:uri_escape_path - Enable/disable URI path escaping (default: true)
	:apply_checksum_header - Include x-amz-content-sha256 header (default: true)



Return Formats
	:list - (default) Returns list of tuples [{"header", "value"}]
	:map - Returns map %{"header" => "value"}
	:req - Returns Req-compatible map %{"header" => ["value"]}

Examples
creds = AWSAuth.Credentials.from_env()

# Returns list of tuples (default)
headers = AWSAuth.sign_authorization_header(creds, "POST", url, "bedrock", headers: %{}, payload: body)
# => [{"authorization", "AWS4-HMAC-SHA256 ..."}, {"x-amz-date", "..."}]

# Auto-detect service and region
headers = AWSAuth.sign_authorization_header(creds, "POST", "https://bedrock-runtime.us-east-1.amazonaws.com/...", nil, payload: body)

# Unsigned payload for streaming
headers = AWSAuth.sign_authorization_header(creds, "PUT", url, "s3", payload: :unsigned)

# Return as map
headers = AWSAuth.sign_authorization_header(
  creds,
  "POST",
  url,
  "bedrock",
  headers: %{"content-type" => "application/json"},
  payload: body,
  return_format: :map
)
# => %{"authorization" => "AWS4-HMAC-SHA256 ...", "x-amz-date" => "..."}

AWSAuth.sign_authorization_header(access_key, secret_key, http_method, url, region, service, headers, payload, request_time, session_token)
access_key: Your AWS Access key
secret_key: Your AWS secret key
http_method: "GET","POST","PUT","DELETE", etc
url: The AWS url you want to sign
region: The AWS name for the region you want to access (i.e. us-east-1). Check here for the region names
service: The AWS service you are trying to access (i.e. s3). Check the url above for names as well.
headers (optional. defaults to Map.new): The headers that will be used in the request. Used for signing the request.
For signing, host is the only one required unless using any other x-amx-* headers.
If host is present here, it will override using the host in the url to attempt signing.
Same goes for the x-amz-content-sha256 headers
If only the host and x-amz-content-sha256 headers are needed, then you don't have to supply it and the host from the url will be used and
the payload will be hashed to get the x-amz-content-sha256 header.
payload (optional. defaults to ""): The contents of the payload if there is one.
request_time (optional): NaiveDateTime for the request timestamp. Defaults to current time.
session_token (optional. defaults to nil): AWS session token for temporary credentials (from STS).
When provided, adds the X-Amz-Security-Token header to the signed request.

  



  
    
      
    
    
      sign_authorization_header(access_key, secret_key, http_method, url, region, service, headers)



        
          
        

    

  


  


  



  
    
      
    
    
      sign_authorization_header(access_key, secret_key, http_method, url, region, service, headers, payload)



        
          
        

    

  


  


  



  
    
      
    
    
      sign_authorization_header(access_key, secret_key, http_method, url, region, service, headers, payload, request_time)



        
          
        

    

  


  


  



  
    
      
    
    
      sign_authorization_header(access_key, secret_key, http_method, url, region, service, headers, payload, request_time, session_token)



        
          
        

    

  


  


  



  
    
      
    
    
      sign_authorization_header(access_key, secret_key, http_method, url, region, service, headers, payload, request_time, session_token, opts)



        
          
        

    

  


  


  



    

  
    
      
    
    
      sign_url(creds, http_method, url, service, opts \\ [])



        
          
        

    

  


  

Signs a URL using AWS Signature V4 with a Credentials struct.
This is a convenience overload that accepts an AWSAuth.Credentials struct.
Parameters
	credentials - AWSAuth.Credentials struct with AWS credentials
	http_method - HTTP method as string ("GET", "POST", etc)
	url - The AWS URL to sign
	service - AWS service name (e.g., "s3", "bedrock"). Can be nil to auto-detect from URL.
	opts - Keyword list of options:	:headers - Request headers (default: %{})
	:payload - Request body (default: "") or :unsigned for streaming
	:timestamp - NaiveDateTime for signing (default: current time)
	:region - Override region from credentials (auto-detected from URL if not provided)
	:expires_in - Expiration time in seconds for presigned URLs (default: 900, max: 604800)
	:uri_escape_path - Enable/disable URI path escaping (default: true)
	:unsigned_headers - List of header names to exclude from signing (default: [])



Examples
creds = AWSAuth.Credentials.from_env()
signed_url = AWSAuth.sign_url(creds, "GET", url, "s3")

# Auto-detect service and region from URL
signed_url = AWSAuth.sign_url(creds, "GET", "https://s3.us-west-2.amazonaws.com/bucket/key", nil)

# Custom expiration (1 hour)
signed_url = AWSAuth.sign_url(creds, "GET", url, "s3", expires_in: 3600)

AWSAuth.sign_url(access_key, secret_key, http_method, url, region, service, headers, request_time, payload, session_token)
access_key: Your AWS Access key
secret_key: Your AWS secret key
http_method: "GET","POST","PUT","DELETE", etc
url: The AWS url you want to sign
region: The AWS name for the region you want to access (i.e. us-east-1). Check here for the region names
service: The AWS service you are trying to access (i.e. s3). Check the url above for names as well.
headers (optional. defaults to Map.new): The headers that will be used in the request. Used for signing the request.
For signing, host is the only one required unless using any other x-amx-* headers.
If host is present here, it will override using the host in the url to attempt signing.
If only the host is needed, then you don't have to supply it and the host from the url will be used.
request_time (optional): NaiveDateTime for the request timestamp. Defaults to current time.
payload (optional. defaults to ""): The contents of the payload if there is one.
session_token (optional. defaults to nil): AWS session token for temporary credentials (from STS).
When provided, adds the X-Amz-Security-Token header to the signed request.

      sign_url(access_key, secret_key, http_method, url, region, service, headers, request_time, payload)



        
          
        

    

  


  


  



  
    
      
    
    
      sign_url(access_key, secret_key, http_method, url, region, service, headers, request_time, payload, session_token)



        
          
        

    

  


  


  



  
    
      
    
    
      sign_url(access_key, secret_key, http_method, url, region, service, headers, request_time, payload, session_token, opts)

AWS credentials struct for authentication.
This struct holds AWS credentials for signing requests. It can be created
manually or loaded from environment variables.
Fields
	:access_key_id - AWS Access Key ID (required)
	:secret_access_key - AWS Secret Access Key (required)
	:session_token - AWS Session Token for temporary credentials (optional)
	:region - AWS region (optional, defaults to "us-east-1")

Examples
# Create credentials manually
creds = %AWSAuth.Credentials{
  access_key_id: "AKIAIOSFODNN7EXAMPLE",
  secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  region: "us-east-1"
}

# Load from environment variables
creds = AWSAuth.Credentials.from_env()

# Load from environment with fallback
creds = AWSAuth.Credentials.from_env() || %AWSAuth.Credentials{
  access_key_id: "...",
  secret_access_key: "..."
}
Environment Variables
from_env/0 reads the following environment variables:
	AWS_ACCESS_KEY_ID - Access key ID
	AWS_SECRET_ACCESS_KEY - Secret access key
	AWS_SESSION_TOKEN - Session token (optional)
	AWS_REGION or AWS_DEFAULT_REGION - Region (optional)

    Types
  


    
      
        t()

      


    





  
    Functions
  


    
      
        from_env()

      


        Creates credentials from environment variables.



    


    
      
        from_map(map_or_keyword)

      


        Creates credentials from a keyword list or map.

          @type t() :: %AWSAuth.Credentials{
  access_key_id: String.t() | nil,
  region: String.t() | nil,
  secret_access_key: String.t() | nil,
  session_token: String.t() | nil
}

          @spec from_env() :: t() | nil


      


Creates credentials from environment variables.
Returns nil if required environment variables are not set.
Examples
# With environment variables set
System.put_env("AWS_ACCESS_KEY_ID", "AKIAIOSFODNN7EXAMPLE")
System.put_env("AWS_SECRET_ACCESS_KEY", "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY")
System.put_env("AWS_REGION", "us-west-2")

creds = AWSAuth.Credentials.from_env()
# => %AWSAuth.Credentials{
#      access_key_id: "AKIAIOSFODNN7EXAMPLE",
#      secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
#      region: "us-west-2",
#      session_token: nil
#    }

# Without required environment variables
System.delete_env("AWS_ACCESS_KEY_ID")
AWSAuth.Credentials.from_env()
# => nil

  



  
    
      
    
    
      from_map(map_or_keyword)



        
          
        

    

  


  

      

          @spec from_map(map() | keyword()) :: t()


      


Creates credentials from a keyword list or map.
Examples
AWSAuth.Credentials.from_map(%{
  access_key_id: "AKIAIOSFODNN7EXAMPLE",
  secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  region: "us-east-1"
})

AWS Signature Version 4 signing for event stream messages.
Some AWS APIs accept a stream of events from the client over a single
connection (for example Amazon Transcribe streaming and Bedrock
bidirectional streaming). Each event is individually signed, and the
signatures form a chain: the first event is signed using the seed
signature from the initial HTTP request's Authorization header, and each
subsequent event is signed using the signature of the event before it.
The signature for an event is computed as:
string_to_sign =
  "AWS4-HMAC-SHA256-PAYLOAD" <> "\n" <>
  <amz_date>                 <> "\n" <>   # YYYYMMDDTHHMMSSZ
  <credential_scope>         <> "\n" <>   # YYYYMMDD/region/service/aws4_request
  <prior_signature>          <> "\n" <>   # hex
  sha256_hex(<header_bytes>) <> "\n" <>
  sha256_hex(<payload>)

signature = hmac_sha256(signing_key, string_to_sign)
High-level usage
sign_message/6 does the whole thing — encode the :date header, sign,
and frame a wire-ready event stream message. Chain the returned signature
into the next call:
seed = "..." # signature from the initial request's Authorization header
creds = AWSAuth.Credentials.from_env() |> Map.put(:region, "us-east-1")

{frame1, sig1} =
  AWSAuth.EventStream.sign_message(creds, "transcribe", seed, payload1, time1)

{frame2, sig2} =
  AWSAuth.EventStream.sign_message(creds, "transcribe", sig1, payload2, time2)
Low-level usage
sign_event/7 mirrors the canonical algorithm exactly (and matches the
aws_signature Erlang library's sign_v4_event/7) when you want to control
the encoded header bytes yourself.

        encode_byte_array_header(name, value)

      


        Encodes a byte-array header (value type 6) into the AWS event stream binary
format. The value is length-prefixed with a 16-bit big-endian length.



    


    
      
        encode_message(headers, payload)

      


        Frames serialized headers and a payload into a complete event stream
message, including the prelude, prelude CRC, and message CRC.



    


    
      
        encode_string_header(name, value)

      


        Encodes a string header (value type 7) into the AWS event stream binary
format. The value is length-prefixed with a 16-bit big-endian length.



    


    
      
        encode_timestamp_header(name, request_time)

      


        Encodes a timestamp header (value type 8) into the AWS event stream binary
format. The value is the milliseconds since the Unix epoch.



    


    
      
        sign_event(credentials, service, prior_signature, header_bytes, payload, request_time, opts \\ [])

      


        Signs a single event and returns the chunk signature as a lowercase hex string.



    


    
      
        sign_message(credentials, service, prior_signature, payload, request_time, opts \\ [])

      


        Signs an event and frames it as a wire-ready event stream message.

      encode_byte_array_header(name, value)



        
          
        

    

  


  

Encodes a byte-array header (value type 6) into the AWS event stream binary
format. The value is length-prefixed with a 16-bit big-endian length.

  



  
    
      
    
    
      encode_message(headers, payload)



        
          
        

    

  


  

Frames serialized headers and a payload into a complete event stream
message, including the prelude, prelude CRC, and message CRC.
Wire format
[total length      :: big-32]
[headers length    :: big-32]
[prelude CRC32      :: big-32]   # crc32 of the 8 prelude bytes above
[headers bytes]
[payload bytes]
[message CRC32      :: big-32]   # crc32 of everything before it

  



  
    
      
    
    
      encode_string_header(name, value)



        
          
        

    

  


  

Encodes a string header (value type 7) into the AWS event stream binary
format. The value is length-prefixed with a 16-bit big-endian length.
Used for application message headers such as :content-type, :event-type,
and :message-type on the events some services expect (e.g. Bedrock
bidirectional streaming input chunks).

  



  
    
      
    
    
      encode_timestamp_header(name, request_time)



        
          
        

    

  


  

Encodes a timestamp header (value type 8) into the AWS event stream binary
format. The value is the milliseconds since the Unix epoch.
request_time may be a NaiveDateTime (interpreted as UTC) or a DateTime.

  



    

  
    
      
    
    
      sign_event(credentials, service, prior_signature, header_bytes, payload, request_time, opts \\ [])



        
          
        

    

  


  

Signs a single event and returns the chunk signature as a lowercase hex string.
This is the low-level entry point. header_bytes must be the event's headers
already serialized in the AWS event stream binary format (typically just the
:date header — see encode_timestamp_header/2), and payload is the raw
event payload.
Options:
	:region - overrides the region from credentials
	:raw - when true, returns the raw 32-byte signature instead of hex

Example
iex> creds = %AWSAuth.Credentials{
...>   access_key_id: "AKIDEXAMPLE",
...>   secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
...>   region: "us-east-1"
...> }
iex> header_bytes = <<5, 58, 100, 97, 116, 101, 8, 0, 0, 1, 137, 171, 187, 255, 224>>
iex> AWSAuth.EventStream.sign_event(
...>   creds,
...>   "transcribe",
...>   "ce2704cf5f348fd66f179d5883162f223c30b3fb8213fb1bc097bf2ecd34b1b5",
...>   header_bytes,
...>   "",
...>   ~N[2023-07-31 11:36:12]
...> )
"29ef82c39850abdcc65f9d6046f3e437e385112b80b7f17b31ba33a7da3cc8af"

  



    

  
    
      
    
    
      sign_message(credentials, service, prior_signature, payload, request_time, opts \\ [])



        
          
        

    

  


  

Signs an event and frames it as a wire-ready event stream message.
Encodes the :date header from request_time, signs the message (using the
encoded :date header as the signed headers and payload as the body),
then builds the full event stream frame containing the :date and
:chunk-signature headers followed by payload.
Returns {frame, signature_hex}. Pass signature_hex as the
prior_signature for the next event in the stream.
Accepts the same options as sign_event/7 (except :raw).

Req plugin for AWS Signature V4 authentication.
This module provides seamless integration between ex_aws_auth and the
Req HTTP client library. It automatically signs requests with AWS Signature V4
without requiring manual header manipulation.
Usage
# Load credentials from environment
creds = AWSAuth.Credentials.from_env()

# Attach to a Req request
Req.new(url: url, method: :post, body: body)
|> AWSAuth.Req.attach(credentials: creds, service: "bedrock")
|> Req.request()

# Or create a reusable client
client =
  Req.new(base_url: "https://bedrock-runtime.us-east-1.amazonaws.com")
  |> AWSAuth.Req.attach(credentials: creds, service: "bedrock")

# Make multiple requests
Req.post!(client, url: "/model/my-model/invoke", json: params)
Options
	:credentials - (required) AWSAuth.Credentials struct or keyword list with:
	:access_key_id - AWS Access Key ID
	:secret_access_key - AWS Secret Access Key
	:session_token - AWS Session Token (optional)
	:region - AWS region (optional, defaults to "us-east-1")


	:service - (required) AWS service name (e.g., "s3", "bedrock", "lambda")

	:region - AWS region (optional, overrides region from credentials)


How it Works
The plugin adds a request step that:
	Extracts the request method, URL, headers, and body
	Normalizes Req's list-valued headers to string values for signing
	Signs the request using AWS Signature V4
	Returns headers in Req's expected format (%{key => [value]})

Examples
# Basic usage with environment credentials
creds = AWSAuth.Credentials.from_env()

response =
  Req.new()
  |> AWSAuth.Req.attach(credentials: creds, service: "s3")
  |> Req.get!(url: "https://my-bucket.s3.amazonaws.com/object")

# With explicit credentials
creds = %AWSAuth.Credentials{
  access_key_id: "AKIAIOSFODNN7EXAMPLE",
  secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  region: "us-west-2"
}

response =
  Req.new(url: url, method: :post, json: data)
  |> AWSAuth.Req.attach(credentials: creds, service: "bedrock")
  |> Req.request!()

# With session token (STS temporary credentials)
creds = %AWSAuth.Credentials{
  access_key_id: "ASIAIOSFODNN7EXAMPLE",
  secret_access_key: "wJalrXUtnFEMI/K7MDENG/bPxRfiCYEXAMPLEKEY",
  session_token: "FwoGZXIvYXdzEBYaDHhBTEMPLESessionToken123",
  region: "us-east-1"
}

response =
  Req.new()
  |> AWSAuth.Req.attach(credentials: creds, service: "lambda")
  |> Req.post!(url: lambda_url, json: payload)

# Override region
response =
  Req.new()
  |> AWSAuth.Req.attach(
    credentials: creds,
    service: "s3",
    region: "eu-west-1"
  )
  |> Req.get!(url: "https://my-bucket.s3.eu-west-1.amazonaws.com/object")

        attach(request, opts)

      


        Attaches AWS Signature V4 signing to a Req request.

          @spec attach(
  Req.Request.t(),
  keyword()
) :: Req.Request.t()


      


Attaches AWS Signature V4 signing to a Req request.
Parameters
	request - A Req.Request struct
	opts - Keyword list of options (see module docs for details)

Returns
A Req.Request with the AWS signing step prepended.
Examples
creds = AWSAuth.Credentials.from_env()

request =
  Req.new(url: url)
  |> AWSAuth.Req.attach(credentials: creds, service: "bedrock")

